

"5.4.BMP"

CALCULATE THE BOUGUER ADMITTANCE AND COHERENCE DUE TO *COMBINED* SURFACE AND BURIED LOADING IN CONTINENTAL *AND* OCEANIC REGIONS.

Mathcad file to calculate the Bouguer admittance and coherence due to combined surface and buried loads in continental and oceanic regions. The method is based on Forsyth (1984) and Ito and Taira (2000) and assumes that the source of the buried loading is located at the base of the crust. The file allows the admittance and coherence to be calculated for any elastic thickness, Te, ratio of surface and buried loading, f, and crustal thickness.

Application to Continental Regions (see Forsyth, 1984)

Define input parameters

Define elastic thickness,Te, average gravity, g, Poisson's ratio, v, Youngs Modulus, E, and the Universal Gravitational constant, G. npts is the number of interpolated values. Must be a power of 2.

$Te := 80km$

$g := 9.81 m \cdot sec^{-2}$

$v := 0.25$

$E := 10^{11} Pa$

$G := 6.67 \cdot 10^{-11} newton \cdot m^{2} \cdot kg^{-2}$

$npts := 1024$

Define density of the load (i.e. the seafloor topography),water and mantle

$\rho\_crust := 2670 kg \cdot m^{-3}$

$\rho\_mantle := 3330 kg \cdot m^{-3}$

Set the ratio of surface to sub-surface loading.

$f := 1$

Set the thickness of the crust.

$continental\_thick := 35km$

Calculate the flexural rigidity, D (flexure).

$$D := \frac{E \cdot (Te)^{3}}{12 \cdot \left(1 - v^{2}\right)}$$

Read in an arbitrary-shaped 2-D topographic load

$M := READPRN("seamount.prn")$

$icountt := rows(M) - 1$

$profile := \left(M_{icountt,0} - M_{0,0}\right) km$

$j := 0 .. icountt$

$a_{j} := M_{j,0} km$

$b_{j} := M_{j,1} m$

Interpolate, take mean out and calculate Fourier Transform of the topography

$$dx := \frac{profile}{npts - 1}$$

$$XKINT := \frac{2 \cdot \pi}{npts \cdot dx}$$

$$XINT := \frac{profile}{npts - 1}$$

$$XKINT = 6.277 \times 10^{-6} \cdot m^{-1}$$

$$XKINT := 6.277 \cdot 10^{-7} m^{-1}$$

$$i := 0 .. npts - 1$$

$$x_i := a_0 + (i \cdot XINT)$$

$$topo_i := linterp\left(a, b, x_i\right)$$

$$mean\_depth := -mean(topo)$$

$$i := 0 .. npts - 1$$

$$y_i := topo_i + mean\_depth$$

Calculate the admittance

$$c := fft(topo)$$

$$k := 0, 1 .. \frac{npts}{2}$$

$$wave_k := XKINT \cdot k$$

$$Phi\_dash_k := \left[\frac{D \cdot (k \cdot XKINT)^4}{g \cdot (\rho\_mantle - \rho\_crust)} + 1\right]^{-1}$$

(See Eq. 5.3)

$$Phi\_3dash_k := \left[\frac{D \cdot (k \cdot XKINT)^4}{g \cdot (\rho\_mantle)} + 1\right]^{-1}$$

(See Eq. 5.37)

Calculate h_t from the observed topography

$$ht_k := \frac{c_k}{\left(1 + \frac{f \cdot Phi\_dash_k \cdot \rho\_crust}{\rho\_mantle - \rho\_crust}\right)}$$

(Substitute Eq. 5.44 for Hb in Eq. 5.50)

Calculate h_b from h_t

$$hb_k := c_k - ht_k$$

$$factor2_k := \frac{\left(\frac{hb_k \cdot hb_k}{Phi\_3dash_k} + ht_k \cdot ht_k \cdot Phi\_dash_k\right)}{\left(hb_k \cdot hb_k + ht_k \cdot ht_k\right)}$$

$$Ztopbot_k := -2 \cdot \pi \cdot G \cdot 10^5 \cdot e^{-(k \cdot XKINT \cdot mean\_depth)} \cdot \rho\_crust \cdot 1000 \cdot e^{-k \cdot XKINT \cdot continental\_thick} \cdot factor2_k$$

(See Eq. 5.43)



Calculate the coherence

$$wt_k := \frac{ht_k \cdot \rho\_crust \cdot Phi\_dash_k}{(\rho\_mantle - \rho\_crust)}$$

(See Eq. 5.34)

$$wb_k := \frac{hb_k \cdot \rho\_mantle}{Phi\_3dash_k \cdot (\rho\_mantle - \rho\_crust)}$$

(See Eq. 5.36)

(See Forsyth Eq. 25)

$$coherence_k := \left[\frac{\left(ht_k \cdot wt_k + hb_k \cdot wb_k\right) \cdot \left(ht_k \cdot wt_k + hb_k \cdot wb_k\right)}{\left(ht_k \cdot ht_k + hb_k \cdot hb_k\right) \cdot \left(wt_k \cdot wt_k + wb_k \cdot wb_k\right)}\right]$$

(Also Eq. 5.51)



Application to Oceanic Regions (see Ito and Taira, 2000)

Define input parameters

Define elastic thickness,Te, for surface and buried loading.

$Te\_surface := 4km$

$Te\_buried := 30km$

Define density of the load (i.e. the seafloor topography),water and mantle.

$\rho\_upper\_crust := 2700kg \cdot m^{-3}$

$\rho\_lower\_crust := 3050kg \cdot m^{-3}$

$\rho\_water := 1000kg \cdot m^{-3}$

$\rho\_mantle := 3300kg \cdot m^{-3}$

Set the ratio of surface to sub-surface loading. Note f=0 corresponds to surface loading only. The *larger f, the larger is the role of buried loading. Also, the mean water depth and the mean thickness of oceanic crust.*

$f := 2$

$\text{mean\_depth} := 4.0\,\text{km}$

$\text{oceanic\_thick} := 21.7\,\text{km}$

Calculate the flexural rigidity for the surface and sub-surface loads.

$$\text{D\_surface} := \frac{E \cdot (\text{Te\_surface})^3}{12 \cdot \left(1 - v^2\right)}$$

$$\text{D\_buried} := \frac{E \cdot (\text{Te\_buried})^3}{12 \cdot \left(1 - v^2\right)}$$

$\text{XKINT} := 6.277 \cdot 10^{-7}\,\text{m}^{-1}$

Calculate the admittance

$k := 0, 1 .. 512$

$\rho\text{\_nought} := \rho\text{\_upper\_crust} - \rho\text{\_water}$

$\text{delta\_}\rho := \rho\text{\_mantle} - \rho\text{\_lower\_crust}$

$$\xi_k := \left[1 + \frac{\text{D\_surface} \cdot (k \cdot \text{XKINT})^4}{\text{delta\_}\rho \cdot g}\right]$$

$\text{wave}_k := \text{XKINT} \cdot k$

$$\phi_k := \left[1 + \frac{\text{D\_buried} \cdot (k \cdot \text{XKINT})^4}{\rho\text{\_nought} \cdot g}\right]$$

$$\beta_k := \frac{\left(\dfrac{\rho\text{\_nought}}{\text{delta\_}\rho \cdot \xi_k} + 1\right)}{\left(\phi_k \cdot \dfrac{\rho\text{\_nought}}{\text{delta\_}\rho} + 1\right)}$$

$$\text{factor}_k := \frac{\left[\left(\dfrac{1}{\xi_k}\right) + \phi_k \cdot f^2 \cdot \left(\beta_k\right)^2\right]}{\left[1 + f^2 \cdot \left(\beta_k\right)^2\right]}$$

$$\text{Ztopbot}_k := -2 \cdot \pi \cdot G \cdot 10^5 \cdot e^{-[k \cdot \text{XKINT} \cdot (\text{mean\_depth} + \text{oceanic\_thick})]} \cdot \rho\text{\_nought} \cdot 1000 \cdot \text{factor}_k$$



Calculate the coherence

$$\text{coherence}_k := \frac{\left[\left(\frac{1}{\xi_k}\right) + \phi_k \cdot f^2 \cdot \left(\beta_k\right)^2\right]^2}{\left[\left[\frac{1}{\left(\xi_k\right)}\right]^2 + \left(\phi_k\right)^2 \cdot f^2 \cdot \left(\beta_k\right)^2\right] \cdot \left[1 + f^2 \cdot \left(\beta_k\right)^2\right]}$$